
local indata LAD\Data
local infile LAD\Data_Cleaning

use "`indata'\LAD_Crowdout.dta", clear

do "`infile'\LAD_Commands_filtering.do"

*Focus on savers within the relevant bandwidth
keep if empinc>=-9000 & empinc<9000

*Non-Unionized RPP Non-Members do not satisfy the Test of Running Variable and are excluded
keep if dues>0
*Focus on RPP members with strictly positive RRSP contributions
keep if penadj>0 & rspcont>0
keep if rspcont<rspdlc & (penadjl+rspcont)<contlimit

gen empincsq=empinc^2
gen empinccb=empinc^3

local covars age agesq female married province_* selfempinc_flag eiinc_flag othinc disab medexp

reg penadj empinc empinc_kink empincsq empinccb `covars'
fitstat, using(pen)
est store rppreg
reg rspcont empinc empinc_kink empincsq empinccb `covars'
fitstat, using(rsp)
est store rspreg

suest rspreg rppreg, cl(id)
nlcom (ratio: [rspreg_mean]_b[empinc_kink]/[rppreg_mean]_b[empinc_kink]), post
test _b[ratio]=-1

exit
